
• There are 5 questions with maximum 7 + 8 + 8 + 9 + 8 = 40 points. Grading:

Chalmers: 3 = 20–26 points, 4 = 27–33 points , 5 = 34–40 points
GU: G = 20–33 points, VG = 34–40 points

• Results: latest approximately 10 days.

• Permitted materials:

– Dictionary

• Please read the following guidelines carefully:

– Read through all Questions before you start working on the answers.

– Begin each Question on a new sheet.

– Write clearly; unreadable = wrong!

– For each part Question, if your solution consists of more than a few lines of Haskell
code, use your common sense to decide whether to include a short comment to
explain your solution.

– You can use any of the standard Haskell functions listed at the back of this exam
document.

– Full points are given to solutions which are short, elegant, and correct. Fewer
points may be given to solutions which are unnecessarily complicated or unstruc-
tured.

– You are encouraged to use the solution to an earlier part of a Question to help
solve a later part — even if you did not succeed in solving the earlier part.

1. (7 points) The standard library function lookup,

lookup :: Eq key => key -> [(key,value)] -> Maybe value

lets you lookup the first value corresponding to a given key in a list of key-value pairs.

(a) (2 points) Define a function that returns all values paired with the given key.

lookupAll :: Eq key => key -> [(key,value)] -> [value]

For 2 points, use a list comprehension. Other solutions will be longer and give at
most 1 point.

(b) (2 points) Use lookupAll to make a simple, non-recursive definition of the standard
lookup function.

(c) (3 points) Define a function update,

update :: Eq key => key -> value -> [(key,value)] -> [(key,value)]

that updates a list of key-value pairs. If the new key already exists in the list, the
value should be replaced, otherwise a new pair should be added to the list. The
order between the pairs in the list does not matter. Examples:

update ’x’ 5 [] == [(’x’,5)]

update ’x’ 5 [(’y’,3)] == [(’y’,3),(’x’,5)]

update ’x’ 5 [(’y’,3),(’x’,2),(’z’,4)] == [(’y’,3),(’x’,5),(’z’,4)]

Solution:

-- (a)

lookupAll key kvs = [v|(k,v)<-kvs,k==key]

-- (b)

lookup key = listToMaybe . lookupAll key

lookup_v2 key kvs = case lookupAll key kvs of

[] -> Nothing

v:_ -> Just v

-- (c)

-- If we assume that k does not appear multiple times in the list

update k v [] = [(k,v)]

update k v ((oldk,oldv):kvs) | oldk==k = (oldk,v) : kvs

| otherwise = (oldk,oldv) : update k v kvs

-- If we update all occurrences of k in the list

update_v2 k v kvs

| null (lookupAll k kvs) = (k,v):kvs

| otherwise = [(k’,if k’==k then v else v’) | (k’,v’)<-kvs]

-- If we remove all previous occurrences of k

update_v3 k v kvs = (k,v):[(k’,v’)|(k’,v’)<-kvs,k’/=k]

2. (8 points) Consider the following data types for representing arithmetic expressions with
multiplication, addition, subtraction and a variable X:

data Expr = X | Num Int | Op BinOp Expr Expr deriving (Eq,Show)

data BinOp = Add | Mul | Subtract deriving (Eq,Show)

ex1 = Op Subtract (Num 100) X -- 100 - X

ex2 = Op Add (Num 100) (Op Mul (Num (-1)) X) -- 100 + (-1)*X

(a) (4 points) Define a function that computes the value of an expression, given the
value of the variable X. Example: eval ex1 25 == 75.

eval :: Expr -> Int -> Int

(b) (4 points) Although this data type can represent subtraction, it is not really needed
since an expression such as, for example, 100 - X can be written as 100 + (-1) * X.

Define a function

removeSub :: Expr -> Expr

which removes all subtraction operators in an expression by replacing them with a
combination of addition and multiplication as in the above example. For example,
removeSub ex1 == ex2.

Your definition should only remove the subtraction operators. It should not at-
tempt to simplify or evaluate the expression in any way.

Solution:

-- (a)

eval X x = x

eval (Num n) x = n

eval (Op op e1 e2) x = evalOp op (eval e1 x) (eval e2 x)

where evalOp Add = (+)

evalOp Mul = (*)

evalOp Subtract = (-)

eval_v2 X x = x

eval_v2 (Num n) x = n

eval_v2 (Op Add e1 e2) x = eval_v2 e1 x + eval_v2 e2 x

eval_v2 (Op Mul e1 e2) x = eval_v2 e1 x * eval_v2 e2 x

eval_v2 (Op Subtract e1 e2) x = eval_v2 e1 x - eval_v2 e2 x

-- (b)

removeSub (Op op e1 e2) = binop op (removeSub e1) (removeSub e2)

where binop Subtract e1 e2 = Op Add e1 (Op Mul (Num (-1)) e2)

binop op e1 e2 = Op op e1 e2

removeSub e = e

removeSub_v2 (Op Add e1 e2) = Op Add (removeSub_v2 e1) (removeSub_v2 e2)

removeSub_v2 (Op Mul e1 e2) = Op Mul (removeSub_v2 e1) (removeSub_v2 e2)

removeSub_v2 (Op Subtract e1 e2) = Op Add r1 (Op Mul (Num (-1)) r2)

where r1 = removeSub_v2 e1

r2 = removeSub_v2 e2

removeSub_v2 e = e

3. (8 points) For each of the following functions, give the most general type, or write ”No
type” if the definition is not type correct in Haskell.

fa x y = not (x && y)

fb (&) x y = not (x & y)

fc x y = (x+y)/2

fd x = [z | y<-x, z<-y]

Solution:

fa :: Bool -> Bool -> Bool

fb :: (a->b->Bool) -> a -> b -> Bool

fc :: Fractional a => a -> a -> a

fd :: [[a]] -> [a]

4. (9 points) Consider the following data type for representing rectangular grids:

data Grid a = Grid [[a]] deriving (Eq,Show)

g1,g2 :: Grid Int -- Example grids

g1 = Grid [[1,2],

[3,4],

[5,6]]

g2 = Grid [[5,3,1],

[6,4,2]]

(a) (2 points) Define a function that a applies a function to every element of a grid.

mapGrid :: (a->b) -> Grid a -> Grid b

(b) (2 points) Define a function that rotates a grid 90 degrees clockwise.

rotateGrid :: Grid a -> Grid a -- Example: rotateGrid g1 == g2

(c) (3 points) Define a QuickCheck test data generator for rectangular grids, including
an instance in the Arbitrary class.

(d) (2 points) Define a property that expresses the fact that rotating a rectangular
grid four times returns the grid you started with.

Hint: The above functions are easier to define by reusing suitable library functions
than by using recursion on lists.

Solution:

-- (a)

mapGrid f = Grid . map (map f) . rows

-- (b)

rotateGrid = Grid . map reverse . transpose . rows

rotateGrid_v2 = Grid . transpose . reverse . rows

rows (Grid rs) = rs -- a common helper function

-- (c)

instance Arbitrary a => Arbitrary (Grid a) where

arbitrary = do height <- choose (1,10)

width <- choose (1,10)

rows <- vectorOf height (vectorOf width arbitrary)

return (Grid rows)

-- (d)

--prop_rotateGrid4 :: Eq a => Grid a -> Bool -- (*)

prop_rotateGrid4 :: Grid Int -> Bool

prop_rotateGrid4 g = (r . r . r . r) g == g

where r = rotateGrid

-- (*) Both types are OK. The property holds for all types of grids, as long

-- as we can test that two grids are equal, but for testing with QuickCheck,

-- we need to choose a particular type.

prop_rotateGrid4_v2 g = iterate rotateGrid g !! 4 == g

5. (8 points) Write a Haskell function

checkHaskellFiles :: IO ()

that examines the Haskell source files in the current directory and reports the ones that
contain lines that are too long. To be more specific:

• For each Haskell file that contains lines that are too long, the function should
report how many lines are too long.

• Lines are too long if they contain more than 78 characters. A Haskell file that does
not contain any lines that are too long is OK.

• The report should not mention Haskell files that are OK. The report should be
empty if all Haskell files are OK.

• Haskell source files have names that end with .hs. Other files can be present, but
they should be ignored.

Example: if the current directory contains the files A.hs (which is OK), B.hs, C.hs,
D.hs and E.pdf (which is not a Haskell source file), the report might look like this:

B.hs: one line is too long

C.hs: 5 lines are too long

D.hs: 12 lines are too long

Hint: In addition to the library functions listed at the back of this exam, the following
library functions might be useful:

-- Functions to output text

putStr, putStrLn :: String -> IO ()

-- listDirectory returns the names of all the files in a directory

listDirectory :: FilePath -> IO [FilePath]

-- readFile reads the contents of a file

readFile :: FilePath -> IO String

-- File names are strings. The name of the current directory is "."

type FilePath = String

Solution:

checkHaskellFiles =

do files <- listDirectory "."

sequence_ [checkHaskellFile f | f<-files,".hs" ‘isSuffixOf‘ f]

where

checkHaskellFile file =

do s <- readFile file

let long = length [l | l<-lines s, length l>78]

case long of

0 -> return ()

1 -> putStrLn (file++": one line is too long")

_ -> putStrLn (file++": "++show long++" lines are too long")

checkHaskellFiles_v2 =

do hfs <- listHaskellFiles "."

hfcs <- zip hfs <$> mapM readFile hfs

putStr (concatMap reportHaskellFile hfcs)

where

listHaskellFiles dir =

do files <- listDirectory "."

return [f | f<-files,".hs" ‘isSuffixOf‘ f]

reportHaskellFile :: (FilePath,String) -> String

reportHaskellFile (filename,sourcecode) =

case length [() | line<-lines sourcecode, length line>78] of

0 -> ""

1 -> filename++": one line is too long\n"

n -> filename++": "++show n++" lines are too long\n"

